
MICROSOFT ACCESS CALENDAR REPORTS
WEEK-AT-A-GLANCE MICROSOFT ACCESS REPORT



OVERVIEW

 Data by itself is of little value. It isn’t 
until we effectively organize, aggregate, 
and display that data gains meaning

 The  quality of  visual presentation of 
data/records helps with our effective 
and rapid interpretation of complex 
information

 Colors, fonts, shapes, positions, and 
other formatting draw our eyes to 
what’s important  

 Familiar patterns assist with 
understanding 

 Thorough yet simple

Charles Joseph Minard's Visualization Of Napoleon's 1812 March

SPECIFICATION

 Create week-at-a-glance calendar similar to Outlook

 Vertical position and height dependent on starting time 
and duration

 Simple with minimal object maintenance

 Data driven flexibility

OTHER SOLUTIONS

 Crystal has a great Calendar Maker 

 Great flexibility (1,000+ lines of code)

 Most code in standard modules

 Richard Rost and many others have 
some great stuff but I couldn’t find the 
method I eventually used.

 Power BI has great potential and has 
made huge breakthroughs in data 
presentation

WEEKLY SCHEDULE BY DOCTOR DESIGN

 DemoWeekly Schedule by Doctor

 Detail section has controls positioned almost 
randomly

 No filters have been used in this demo

 Simple black and white

 Doctor then Week of groupings 

 Detail and WeekOf footer both 6”

WEEKLY SCHEDULE BY DOCTOR CODE

Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)
Dim lngTopMargin As Long
Dim lngOneMinute As Long 'size of one minute in twips
Dim datSchedStart As Date

datSchedStart = #8:00:00 AM#
lngOneMinute = 12  'number of twips in one minute
lngTopMargin = Me.boxTimeLine.Top ' 720 'timeline starts 1/2" down in section
'create detail sections as 'layers'
Me.MoveLayout = False
Me.Patient.Top = lngTopMargin + DateDiff("n", datSchedStart, Me.StartTime) * lngOneMinute
Me.Patient.Height = DateDiff("n", Me.StartTime, Me.EndTime) * lngOneMinute
Me.Patient.Left = DateDiff("d", Me.WeekOf, Me.SchedDate) * 2160

End Sub

LIMITATIONS & OPPORTUNITIES

 Limitations

 Isn’t designed to handle events outside of standard scheduling

 Somewhat small text boxes to display names

 Opportunities

 Dynamically draw time scale 

 Encorporate filtering as needed for date range and doctor

 Additional information could be added

DAILY SCHEDULE ALL DOCTORS DESIGN

 Demo Daily Schedule All Doctors

 Again detail section has controls positioned 
almost randomly

 ReportColumn is horizontal position of doctor 
column (twips)

 Simple black and white

 Grouped by SchedDate

 Detail and SchedDate footer both 7”

DAILY SCHEDULE ALL DOCTORS CODE

Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)
Dim lngTopMargin As Long
Dim lngOneMinute As Long            'size of one minute in twips
Dim datSchedStart As Date
datSchedStart = #8:00:00 AM#
lngOneMinute = 12
lngTopMargin = 1440                 'timeline starts 1" down in section
Me.MoveLayout = False
Me.Patient.Top = lngTopMargin + DateDiff("n", datSchedStart, Me.StartTime) * lngOneMinute
Me.Patient.Height = DateDiff("n", Me.StartTime, Me.EndTime) * lngOneMinute
Me.Patient.Left = Me.ReportColumn 'field in table
Me.Doctor.Left = Me.ReportColumn 'field in table

End Sub

LIMITATIONS & OPPORTUNITIES

 Limitations

 Isn’t designed to handle events outside of standard scheduling

 Somewhat small text boxes to display names

 Opportunities

 Dynamically draw time scale 

 Encorporate filtering as needed for date range and doctor

 Additional information could be added similar to task list in Outlook calendar view

CREW ROTATION SCHEDULE DESIGN

 Demo Crew Rotation Schedule

 Again detail section has controls positioned almost randomly

 Timeline created across top

 Color used for Vessel

 Grouped by Vessel

 Detail and Vessel footer both 0.3”

CREW ROTATION SCHEDULE CODE

Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)
Dim lngDuration As Long 'days of tour
Dim lngStart As Long                       'start date of tour
Dim lngLMarg As Long
Dim dblFactor As Double
'put a line control in your page header that starts 1/1 and goes to 12/31
lngLMarg = Me.boxTimeLine.Left
dblFactor = Me.boxTimeLine.Width / 365
lngStart = DateDiff("d", #1/1/2001#, Me.[Start Date])
lngDuration = DateDiff("d", Me.[Start Date], Me.[End Date])
'set the color of the bar based on a data value
Me.txtName.BackColor = Me.VesselColor
Me.txtName.Width = 10                       'avoid the positioning error
Me.txtName.Left = (lngStart * dblFactor) + lngLMarg
Me.txtName.Width = (lngDuration * dblFactor)
Me.MoveLayout = False

End Sub

DUAL VALUE CHART DESIGN

 Demo Dual Value Chart

 Again detail section has controls positioned 
almost randomly

 Math is used to calculate height and width of 
chart bars

DUAL VALUE CHART CODE
Private Sub Detail_Format(Cancel As Integer, FormatCount As Integer)

Dim intSpaceBetween As Integer 'spacing between bars
Dim intBarHeight As Integer
Dim intSpacePerEmployee As Integer 'used to create bar widths
'calculate the bar height based on the percent of maximum and height of box
intBarHeight = Me.boxBorder.Height * 0.9 * (Me.SalaryGroup / Me.txtMaxSalaryGroup)
intSpaceBetween = 100
'calculate the space of a single employee based on width of box, _

number of employees and spaces between
intSpacePerEmployee = (Me.boxBorder.Width - (intSpaceBetween * _

(Me.txtCountGroups + 2))) / Me.txtCountEmps
'code that overlays all detail sections
Me.MoveLayout = False
'find the top of the salary bar
Me.CountOfSalary.Top = Me.boxBorder.Top + Me.boxBorder.Height - intBarHeight
'set the height of the bar
Me.CountOfSalary.Height = intBarHeight
'set the vertical position of the labels under the bars
Me.SalaryGroup.Top = Me.boxBorder.Top + Me.boxBorder.Height + 60
'set the width of the bar
'set the width prior to the left to avoid error of can't place the control
Me.CountOfSalary.Width = Me.CountOfSalary * intSpacePerEmployee
'set the left of the bar
Me.CountOfSalary.Left = Me.boxBorder.Left + intSpaceBetween * Me.txtCount + _

(Me.txtCountOfSalaryRunSum - Me.CountOfSalary) * intSpacePerEmployee
'set the left of the group labels under bars
Me.SalaryGroup.Left = Me.CountOfSalary.Left
Me.Total_Salary.Top = Me.CountOfSalary.Top + Me.CountOfSalary.Height / 2.5
Me.Total_Salary.Left = Me.CountOfSalary.Left + Me.CountOfSalary.Width / 2 _

- intSpaceBetween * 2
End Sub

CALENDAR STYLE SCHEDULE DESIGN

 Demo Calendar Style Schedule

 Groups by Doctor and WeekOf

 Uses five copies of the same subreport of a single day

 Five text boxes across top of detail calculate dates which are used as Link Master for the daily 
subreports

 Uses Line method in On Print event of detail to border subreports based on tallest date

CALENDAR STYLE SCHEDULE CODE

Private Sub Detail_Print(Cancel As Integer, PrintCount As Integer)
Dim lngHeight As Long 'store the height of tallest control
Dim i As Integer
Dim lngLeft As Long
For i = 1 To 5

'compare heights
If Me("srpt" & i).Height > lngHeight Then

lngHeight = Me("srpt" & i).Height
End If

Next
'add the height of other control
lngHeight = lngHeight + Me.txtDay1.Height
'draw the lines
For i = 0 To 5

lngLeft = i * Me.srpt1.Width
Me.Line (lngLeft, 0)-(lngLeft, lngHeight)

Next
End Sub

LIMITATIONS & OPPORTUNITIES

 Limitations

 Much of the design is hard-code for simplicity sake

 It’s on paper and not dynamic

 Only works where code runs (Print Preview and 
Run Report)

 Doesn’t support odd shapes without a lot of work

 Opportunities

 Might be used for seating charts, cemetery plots, 
org charts, graphs, etc.

 Make more dynamic and use the report’s Print 
method with CurrentX and CurrentY

QUESTIONS - DISCUSSION

